```
> restart;
> # Modified to construct a lower order predictor that is stable...
> ynp1:=a0*y(tn)+a1*y(tn-h)+a2*y(tn-2*h)+
       h*(b0*D(y)(tn)+b1*D(y)(tn-h)+
          b2*D(y)(tn-2*h))+E4*h^4*(D@@4)(y)(theta)/4!;
```

$$ynp1 := a0\,y(tn) + a1\,y(tn-h) + a2\,y(tn-2\,h) + h\,(b0\,\mathrm{D}(y)(tn) + b1\,\mathrm{D}(y)(tn-h) + b2\,\mathrm{D}(y)(tn-2\,h)) + \frac{E4\,h^4\,\mathrm{D}^{(4)}(y)(\theta)}{24} \quad (1)$$

```
> r:=y(tn+h)-ynp1;
```

$$r := y(tn+h) - a0\,y(tn) - a1\,y(tn-h) - a2\,y(tn-2\,h) - h\,(b0\,\mathrm{D}(y)(tn) + b1\,\mathrm{D}(y)(tn-h) + b2\,\mathrm{D}(y)(tn-2\,h)) - \frac{E4\,h^4\,\mathrm{D}^{(4)}(y)(\theta)}{24} \quad (2)$$

```
> eq[0]:=eval(subs(y=(x->1),r));
```

$$eq_0 := 1 - a0 - a1 - a2 \quad (3)$$

```
> for j from 1 to 4 do
     tmp[j]:=eval(subs(y=(x->x^j),r));
     eq[j]:=coeff(tmp[j],h^j);
     print(eq[j]);
  od:
```

$$1 + a1 + 2\,a2 - b0 - b1 - b2$$
$$1 - a1 - 4\,a2 + 2\,b1 + 4\,b2$$
$$1 + a1 + 8\,a2 - 3\,b1 - 12\,b2$$
$$1 - a1 - 16\,a2 + 4\,b1 + 32\,b2 - E4 \quad (4)$$

```
> S1:=solve({seq(eq[k]=0,k=0..4)},{a0,b0,b1,b2,E4});
```

$$S1 := \left\{E4 = 9 - a1,\ a0 = 1 - a1 - a2,\ b0 = \frac{23}{12} + \frac{5\,a1}{12} + \frac{a2}{3},\ b1 = -\frac{4}{3} + \frac{2\,a1}{3} + \frac{4\,a2}{3},\ b2 = \frac{5}{12} - \frac{a1}{12} + \frac{a2}{3}\right\} \quad (5)$$

```
> method:=subs(E4=0,ynp1);
```

$$method := a0\,y(tn) + a1\,y(tn-h) + a2\,y(tn-2\,h) + h\,(b0\,\mathrm{D}(y)(tn) + b1\,\mathrm{D}(y)(tn-h) + b2\,\mathrm{D}(y)(tn-2\,h)) \quad (6)$$

```
> m2:=subs(S1,method);
```

$$m2 := (1 - a1 - a2)\,y(tn) + a1\,y(tn-h) + a2\,y(tn-2\,h) + h\left(\left(\frac{23}{12} + \frac{5\,a1}{12} + \frac{a2}{3}\right)\mathrm{D}(y)(tn) + \left(-\frac{4}{3} + \frac{2\,a1}{3} + \frac{4\,a2}{3}\right)\mathrm{D}(y)(tn-h) + \left(\frac{5}{12} - \frac{a1}{12} + \frac{a2}{3}\right)\mathrm{D}(y)(tn-2\,h)\right) \quad (7)$$

```
> m3:=eval(subs(D(y)=(x->f(x,y(x))),m2));
```

$$m3 := (1 - a1 - a2)\,y(tn) + a1\,y(tn-h) + a2\,y(tn-2\,h) + h\left(\left(\frac{23}{12} + \frac{5\,a1}{12} + \frac{a2}{3}\right)f(tn, y(tn)) + \left(-\frac{4}{3} + \frac{2\,a1}{3} + \frac{4\,a2}{3}\right)f(tn-h, y(tn-h)) + \left(\frac{5}{12}\right.\right. \quad (8)$$

$$-\frac{a1}{12}+\frac{a2}{3}\Big)\,f(tn-2\,h,\,y(tn-2\,h))\Big)$$

```
> f:=(xi,eta)->A*eta;
```

$$f := (\xi, \eta) \mapsto A\cdot\eta \qquad \textbf{(9)}$$

```
> m4:=y(tn+h)=m3;
```

$$m4 := y(tn+h) = (1-a1-a2)\,y(tn) + a1\,y(tn-h) + a2\,y(tn-2\,h) + h\Bigg(\Big(\frac{23}{12} + \frac{5\,a1}{12} + \frac{a2}{3}\Big)\,A\,y(tn) + \Big(-\frac{4}{3} + \frac{2\,a1}{3} + \frac{4\,a2}{3}\Big)\,A\,y(tn-h) + \Big(\frac{5}{12} - \frac{a1}{12} + \frac{a2}{3}\Big)\,A\,y(tn-2\,h)\Bigg) \qquad \textbf{(10)}$$

```
> ceq:=eval(subs(y=(s->rho^s),m4));
```

$$ceq := \rho^{tn+h} = (1-a1-a2)\,\rho^{tn} + a1\,\rho^{tn-h} + a2\,\rho^{tn-2\,h} + h\Bigg(\Big(\frac{23}{12} + \frac{5\,a1}{12} + \frac{a2}{3}\Big)\,A\,\rho^{tn} + \Big(-\frac{4}{3} + \frac{2\,a1}{3} + \frac{4\,a2}{3}\Big)\,A\,\rho^{tn-h} + \Big(\frac{5}{12} - \frac{a1}{12} + \frac{a2}{3}\Big)\,A\,\rho^{tn-2\,h}\Bigg) \qquad \textbf{(11)}$$

```
> ceq2:=subs({a2=1/3,a1=0,tn=1,h=1},ceq);
```

$$ceq2 := \rho^2 = \frac{2\,\rho}{3} + \frac{1}{3\,\rho} + \frac{73\,A\,\rho}{36} - \frac{8\,A}{9} + \frac{19\,A}{36\,\rho} \qquad \textbf{(12)}$$

```
> S2:=solve(ceq2,rho):
> Z1:=subs(A=a+I*b,abs(S2[1])):
> Z2:=subs(A=a+I*b,abs(S2[2])):
> Z3:=subs(A=a+I*b,abs(S2[3])):
> with(plots):
> contourplot(max(Z1,Z2,Z3),a=-2..1,b=-1..1,contours=[1],
    grid=[100,100],filled=true);
```



```
> S1;
```

$$\left\{E4 = 9 - a1,\ a0 = 1 - a1 - a2,\ b0 = \frac{23}{12} + \frac{5\,a1}{12} + \frac{a2}{3},\ b1 = -\frac{4}{3} + \frac{2\,a1}{3} + \frac{4\,a2}{3},\ b2 = \frac{5}{12} - \frac{a1}{12} + \frac{a2}{3}\right\} \quad (13)$$

```
> M1:=subs(S1,a0^2+a1^2+a2^2);
```

$$M1 := (1 - a1 - a2)^2 + a1^2 + a2^2 \quad (14)$$

```
> M2:=subs({a1=x,a2=x},M1);
```

$$M2 := (1 - 2\,x)^2 + 2\,x^2 \quad (15)$$

```
> M3:=diff(M2,x);
```

$$M3 := -4 + 12\,x \quad (16)$$

```
> solve(M3=0,x);
```

$$\frac{1}{3} \quad (17)$$

```
> subs({a1=0,a2=1/3},M1); # still less than 1
```

$$\frac{5}{9} \quad (18)$$

```
> subs({a1=1/3,a2=1/3},M1); # better but also less stable
```

$$\frac{1}{3} \quad (19)$$